
Stata tip 20: Generating histogram bin variables

David A. Harrison
ICNARC, London, UK

david.harrison@icnarc.org

Did you know about twoway histogram gen? (Note the two underscores in the
first gap and only one in the second.) This command is used by histogram to generate
the variables that are plotted. It is undocumented in the manuals but explained in the
online help. The command can be used directly to save these variables, enabling more
complex manipulation of histograms and production of other graphs or tables.

Consider the S&P 500 historical data that are used as an example for [R] histogram:

. use http://www.stata-press.com/data/r9/sp500
(S&P 500)

. histogram volume, percent start(4000) width(1000)
(bin=20, start=4000, width=1000)

(output omitted )

To display only the central part of this histogram from 8,000 to 16,000, we could
use if, but this will change the height of the bars, as data outside the range 8,000 to
16,000 will be ignored completely. To restrict the range without altering the bars, we
use twoway histogram gen to save the histogram and only plot the section of interest:

. twoway__histogram_gen volume, percent start(4000) width(1000) gen(h x)

. twoway bar h x if inrange(x,8000,16000), barwidth(1000) bstyle(histogram)

The start() and width() options above specified cutpoints that included 8,000 and
16,000. We could, alternatively, use the default cutpoints:

. twoway__histogram_gen volume if inrange(volume,8000,16000), display
(bin=14, start=8117, width=525.08571)

. local m = r(start)

. local w = r(width)

. summarize volume, meanonly

. local s = ‘m’ - ‘w’ * ceil((‘m’ - r(min))/‘w’)

. twoway__histogram_gen volume, percent start(‘s’) width(‘w’) gen(h x, replace)

. twoway bar h x if inrange(x,8000,16000), barwidth(‘w’) bstyle(histogram)
(output omitted )

Other uses of twoway histogram gen include the following:

• Overlaying or mirroring two histograms

. use http://www.stata-press.com/data/r9/bplong, clear
(fictional blood-pressure data)

. twoway__histogram_gen bp if sex == 0, frac start(125) w(5) gen(h1 x1)

. twoway__histogram_gen bp if sex == 1, frac start(125) w(5) gen(h2 x2)

. twoway (bar h1 x1, barw(5) bc(gs11))
> (bar h2 x2, barw(5) blc(black) bfc(none)),
> legend(order(1 "Male" 2 "Female"))

(output omitted )

. qui replace h2 = -h2

. twoway (bar h1 x1, barw(5)) (bar h2 x2, barw(5)),
> yla(-.2 ".2" -.1 ".1" 0 .1 .2) legend(order(1 "Male" 2 "Female"))

(output omitted )

• Changing the scale, for example, to plot density on a square-root scale

. twoway__histogram_gen bp, start(125) width(5) gen(h x)

. qui gen hsqrt = sqrt(h)

. twoway bar hsqrt x, barw(5) bstyle(histogram) ytitle(Density)
> ylabel(0 .05 ".0025" .1 ".01" .15 ".0225" .2 ".04")

(output omitted )

• Plotting the differences between observed and expected frequencies

. twoway__histogram_gen bp, freq start(125) w(5) gen(h x, replace)

. qui summarize bp

. qui gen diff = h - r(N) * (norm((x + 2.5 - r(mean))/r(sd)) -
> norm((x - 2.5 - r(mean))/r(sd)))

. twoway bar diff x, barw(5) yti("Observed - expected frequency")
(output omitted )

There are also two similar commands: twoway function gen to generate functions
and twoway kdensity gen to generate kernel densities.




